
2 [bookmark: _Toc140667222][bookmark: _Toc141177416]Corrections required
2.1 Correction to function fl_NBIOT_CheckSupportedBands in f_TC_22_4_13
	Function name
	fl_NBIOT_CheckSupportedBands

	Reason for change
	1. The current function, fl_NBIOT_CheckSupportedBands, can let a non-conformant UE pass the testcase when the first reported band matches the first band configured via PICS (eg. UE supports band 3 only while PICS bands are set as 3, 4, 5, 12).

2. The current function, fl_NBIOT_CheckSupportedBands, will also not evaluate the correct PICS band because of one of the variables, v_MaxBandsReported, reporting the length of the list and not the maximum band reported and can cause the TC to not report a failure for a band that should have triggered a failure.


	Summary of change
	1. Rewritten the function to not use for loops inbedded in for loops for clarity. Use of comparison of two bitstrings (v_SuppBandsFromPICS and v_SuppBandsReported) to determine if a mismatch between reported bands and configured bands from PICS. Please see below for details.

Creation of a new variable named v_MaxBandReported to hold an integer of the final (largest) band reported by the UE.

Creation of a new variable named v_NumBandsReported to hold the number of bands reported by the UE.

Creation of a new variable v_NumSupportedBands to hold the number of supported bands from tsc_NBIOT_SupportedBands

Additionally, a check on if the UE did report any bands at all is implemented.

	TTCN module
	NBIOT_RRC_ControlPlane.ttcn

	MCC160 Comment
	



Before
function fl_NBIOT_CheckSupportedBands(SupportedBandList_NB_r13 p_SupportedBandList)
  {
	var integer i;
    var integer j;
    var integer v_Band_r13;
    var integer v_BitStringIndex;
    var bitstring v_SuppBandsFromPICS := f_DeriveSuppNBIoTBandsFromPics ();
    var integer v_MaxBandsReported := lengthof(p_SupportedBandList);
    var bitstring v_SuppBandsReported := int2bit(0, v_MaxBandsReported); // 1... maxbands
    // There may be more bands reported than there are PICS defined for bands
    
    for (i:= 0; i < lengthof(p_SupportedBandList); i:= i+1) {
      if (tsc_NBIOT_SupportedBands[i] == '1'B) { // Don't check anything if band not supported @sic R5s170290 sic@
        if (v_SuppBandsFromPICS[i] =='1'B) { // PICS for band i
          for (j := 0; j < v_MaxBandsReported; j := j + 1) { // check against all reported bands
            v_Band_r13 := p_SupportedBandList[j].band_r13;
            v_BitStringIndex := v_Band_r13 - 1;
            if (i == v_BitStringIndex) {
              if (v_SuppBandsReported[v_BitStringIndex] == '1'B) {
                f_SetVerdict(fail, __FILE__, __LINE__, "support of band nr. " & int2str(v_Band_r13) &  " is reported more than once"); // error: duplicate found
              } else {
                v_SuppBandsReported[v_BitStringIndex] := '1'B;
              }
            }
            else {// do nothing
            }
          }
          if (v_SuppBandsReported[i] != '1'B) { // PICS set,  but expected band was not in the list
            f_SetVerdict(fail, __FILE__, __LINE__, "expected (per PICS) band nr. " & int2str(i + 1) &  " is not reported"); // error: PICS set, but expected band not reported
          }
			}
        else { // PICS is not set => non support
          for (j := 0; j < v_MaxBandsReported; j := j + 1) { // check against all reported bands
            v_Band_r13 := p_SupportedBandList[j].band_r13;
            v_BitStringIndex := v_Band_r13 - 1;
            if (i == v_BitStringIndex) {
              f_SetVerdict(fail, __FILE__, __LINE__, "PICS does not match reported capability for band nr. " & int2str(v_Band_r13)); // PICS mismatch
            }
            else { // do nothing
            }
          }
        }
      } // if band not supported
		}
	}

After
  function fl_NBIOT_CheckSupportedBands(SupportedBandList_NB_r13 p_SupportedBandList)
  {
      
    if (lengthof(p_SupportedBandList) == 0){
        f_SetVerdict(fail, __FILE__, __LINE__, "No bands were reported by the UE"); // error: No bands reported by the UE
    }//WA#117600
      
    var integer i;
    var integer v_Band_r13;
    var integer v_BitStringIndex;
    var bitstring v_SuppBandsFromPICS := f_DeriveSuppNBIoTBandsFromPics (); 
    var integer v_NumBandsReported := lengthof(p_SupportedBandList); // Number of bands reported by UE // There may be more bands reported than there are PICS defined for bands
    var integer v_MaxBandReported := p_SupportedBandList[v_NumBandsReported - 1].band_r13; // Final band reported, to compute the length of bitstring
    var integer v_NBIOT_NumSupportedBands := lengthof(tsc_NBIOT_SupportedBands); // The number of supported bands to test
    var bitstring v_SuppBandsReported := int2bit(0, v_MaxBandReported);
      
    for (i := 0; i < v_NumBandsReported; i := i + 1) { // setup of bitstring of UE's reported bands to compare with PICS bitstring bands
        v_Band_r13 := p_SupportedBandList[i].band_r13; 
        v_BitStringIndex := v_Band_r13 - 1;
        if (v_SuppBandsReported[v_BitStringIndex] == '1'B) { // if UE reports the same band twice
            f_SetVerdict(fail, __FILE__, __LINE__, "support of band " & int2str(v_Band_r13) &  " is reported more than once"); // error: duplicate found
        }
        v_SuppBandsReported[v_BitStringIndex] := '1'B
    }
      
      //if v_MaxBandReported is less than v_NBIOT_NumSupportedBands, append 0 bits to v_SuppBandsReported bitstring until the same size
      if (v_MaxBandReported < v_NBIOT_NumSupportedBands){
          for(i := v_MaxBandReported; i < v_NBIOT_NumSupportedBands; i := i + 1) {
              v_SuppBandsReported[i] := '0'B;
          }
      }

        // For every supported band check band set in PICS against band reported
    for (i := 0; i < v_NBIOT_NumSupportedBands; i:= i+1) {
         if (v_SuppBandsFromPICS[i] != v_SuppBandsReported[i]) { // if UE reported bands and PICS set bands don't match
              f_SetVerdict(fail, __FILE__, __LINE__, "PICS and reported band list mismatch: PICS and UE eband: " & int2str(i+1) & " is mismatched. UE reported: " & bit2str(v_SuppBandsReported[i]) & ", while PICS was set to: " & bit2str(v_SuppBandsFromPICS[i])) // error: PICS and UE band list do not match
         }
    }  
  }

